Modeling Sensors and other Physical Systems with SPICE

(Part 1)

Dr. José Ernesto Rayas-Sánchez

1

Simulation of Sensors and Transducers

- Sensors, transducers, and other multi-physical components can be simulated in SPICE
- SPICE models for those components are based on:
  - Analytical functions
    - Multidimensional vector functions
    - Nonlinear systems of equations
    - Systems of nonlinear differential equations
  - Measurements

Dr. J. E. Rayas-Sánchez

2

Multidimensional Vector Functions



$\boldsymbol{x}$ Vector of physical variables (temperature, humidity, pressure, volumetric flow, etc.)

$\boldsymbol{y}$ Vector of electrical outputs (voltage, current, charge, resistance, etc.)

$$\boldsymbol{y} = \boldsymbol{f}(\boldsymbol{x}) \qquad \boldsymbol{f} : \Re^n \rightarrow \Re^m$$

It is implemented in SPICE by using controlled sources and independent sources (*R. Saleh and A. Yang, 1993*)

Implementing $\boldsymbol{y} = \boldsymbol{f}(\boldsymbol{x})$ with SPICE

1) Implement each physical variable, x_i, …, x_n with a DC voltage source, isolated from the rest of the circuit
2) Implement each constant or any other parameter using also an isolated DC voltage source
3) Use a controlled source to implement each function f_1, …, f_m, using linear controlled sources, polynomial controlled sources, or nonlinear controlled sources

Linear Controlled Sources

- Types: $i = gv_c$, $v = ev_c$, $i = fi_c$, $v = hi_c$
  where g, e, f, y h are real constants, v_c and i_c are the controlling signals
- Syntax

```
GXXXXXXX N+ N- NC+ NC- VALUE
EXXXXXXX N+ N- NC+ NC- VALUE
FXXXXXXX N+ N- VNAM VALUE
HXXXXXXX N+ N- VNAM VALUE
```

(VNAM is the name of the voltage source by which the controlling current is flowing)

Dr. J. E. Rayas-Sánchez 5

Examples of Linear Controlled Sources

$\text{E1} = v_{5-3} = 3v_{11}$

```
E1 5 3 11 0 3.0
```

$\text{G2} = i_{1-0} = 0.5\times 10^{-3} v_5$

```
G2 1 0 5 0 0.5MMHO
```

$\text{Hfvcc} = v_{8-15} = 900 i_{Vmedir}$

```
Hfvcc 8 15 Vmedir 0.9K
```

$\text{F1} = i_{7-51} = 10 i_{Vsens}$

```
F1 7 51 VSENS 10
```

Dr. J. E. Rayas-Sánchez 6

Polynomial Controlled Sources

- Types: $i = g(v_c)$, $v = e(v_c)$, $i = f(i_c)$, $v = h(i_c)$
  where *g*, *e*, *f*, y *h* are polynomial functions of one or more variables, and v_c, i_c are the controlling signals
- Syntax

```
GXXXXXXX N+ N- <POLY(ND)> NC1+ NC1- ... p0 <p1 ...> <IC=...>
EXXXXXXX N+ N- <POLY(ND)> NC1+ NC1- ... p0 <p1 ...> <IC=...>
FXXXXXXX N+ N- <POLY(ND)> VN1 <VN2 ...> p0 <p1 ...> <IC=...>
HXXXXXXX N+ N- <POLY(ND)> VN1 <VN2 ...> p0 <p1 ...> <IC=...>
```

(VN1, VN2, ... are the names of the voltage sources through which the controlling currents are flowing, ND is the polynomial dimension, IC are the initial conditions)

Polynomial Controlled Sources (cont.)

Interpretation of polynomial coefficients (p0, p1, ...)

- If ND = 1

y = p0 + (p1*a) + (p2*a**2) + (p3*a**3) + (p4*a**4) + (p5*a**5) + ...

- If ND = 2

y = p0 + (p1*a) + (p2*b) + (p3*a**2) + (p4*a*b) + (p5*b**2)
+ (p6*a**3) + (p7*a**2*b) + (p8*a*b**2) + (p9*b**3) + ...

- If ND = 3

y = p0 + (p1*a) + (p2*b) + (p3*c) + (p4*a**2) + (p5*a*b)
+ (p6*a*c) + (p7*b**2) + (p8*b*c) + (p9*c**2) + (p10*a**3)
+ (p11*a**2*b) + (p12*a**2*c) + (p13*a*b**2) + (p14*a*b*c)
+ (p15*a*c**2) + (p16*b**3) + (p17*b**2*c) + (p18*b*c**2)
+ (p19*c**3) + (p20*a**4) + ...

Examples of Polynomial Controlled Sources

$\text{GR} = i_{17-3} = 10^{-3}(v_{10-3} + 1.5v_{10-3}^2)$

```
GR 17 3 10 3 0 1M 1.5M
```

$\text{EX} = v_{18} = v_{13} + v_{15} + v_{17}$

```
EX 18 0 POLY(3) 13 0 15 0 17 0 0 1 1 1
```

$\text{F1} = i_{12-10} = 1\text{mA} - 1.5\text{mA}i_{Vcc}$

```
F1 12 10 VCC 1MA -1.5M
```

$\text{HXY} = v_{13-20} = 500 i_{VIN1} i_{VIN2}$

```
HXY 13 20 POLY(2) VIN1 VIN2 0 0 0 0 500
```


Nonlinear Controlled Sources

- Type: $y = b(v_c, i_c)$

  where y can be a voltage or a current, b is an arbitrary function of the DC values of the controlling signals v_c and i_c
- Syntax

  ```
  BXXXXXXX N+ N- <I=EXPR> <V=EXPR>
  ```
- Available functions and operators for the expression:

abs	asinh	cosh	sin			
acos	atan	exp	sinh	+	-	*
acosh	atanh	ln	sqrt	/	^	unary -
asin	cos	log	tan			

Examples of Nonlinear Controlled Sources

$i_2 = \cos(v_1) + \sin(v_3)$

```
B1 0 2 I=cos(v(1))+sin(v(3))
```

$v_{2-3} = \ln(\cos(\log(v_{1-2}^2))) + v_3 v_1$

```
B2 2 3 V=ln(cos(log(v(1,2)^2)))+v(3)*v(1)
```

$v_5 = 7e^{\pi i_{vd}}$

```
B3 5 0 V=7*exp(pi*i(Vd))
```


Ejemplo: Sensor de Presión Piezoresistivo

Simular un transductor piezoeléctrico de dos terminales cuya resistencia eléctrica r depende de su estrés mecánico σ (presión) y de su temperatura T, según la ecuación:

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

r_0 es la resistencia del transductor a 0 grados centígrados y 0 KPa de presión (r_0 = 10KΩ)

α y k son constantes conocidas ($\alpha = 5\times10^{-3}$ y $k = 0.006$)

Sensor de Presión Piezoresistivo

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$





$$i = \frac{V_1 - V_2}{V(r_0)\left(1 + V(\alpha)V(T) + \dfrac{2V(k)V(\sigma)}{1 - V(k)V(\sigma)}\right)}$$

Dr. J. E. Rayas-Sánchez 13

Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

($r_0 = 10\text{K}\Omega$, $\alpha = 5\times10^{-3}$, $k = 0.006$)

probando la ecuación...

```
Strain Transducer Equation
*
vk k 0 DC 0.006
vT T 0 DC 0
valfa alfa 0 DC 5e-3
vsigma sigma 0 DC 0
vr0 r0 0 DC 10K
bp 0 r i=v(r0)*(1+v(alfa)*v(T)+2*v(k)*v(sigma)/(1-v(k)*v(sigma)))
RL r 0 1
.control
DC vsigma 0 100 0.1
plot v(r)
DC vT -40 120 0.1
plot v(r)
DC vsigma 0 100 1 vT -40 120 20
plot v(r)
.endc
.end
```

Dr. J. E. Rayas-Sánchez 14

Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

r *vs.* σ con $T = 0$ °C



Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

r *vs.* T con $\sigma = 0$ KPa



Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1+\alpha T + \frac{2k\sigma}{1-k\sigma}\right)$$

r *vs.* σ con T : –40 a 120 °C



Sensor de Presión Piezoresistivo (cont)

Plotting in WinSpice with suitable text labels:

```
.control
destroy all
DC vsigma 0 100 0.1
r = v(r)
plot r xunits KPa yunits Ohms xlabel sigma
DC vT -40 120 0.1
r = v(r)
plot r xunits Celsius yunits Ohms xlabel T
DC vsigma 0 100 1 vT -40 120 20
r = v(r)
plot r xunits KPa yunits Ohms xlabel sigma
.endc
```

(only for WinSpice version 2004 or newer)

Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

r *vs.* σ con $T = 0$ °C



Dr. J. E. Rayas-Sánchez 19

Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

r *vs.* T con $\sigma = 0$ KPa



Dr. J. E. Rayas-Sánchez 20

Sensor de Presión Piezoresistivo (cont)

$$r = r_0\left(1+\alpha T+\frac{2k\sigma}{1-k\sigma}\right)$$

r *vs.* σ con T: –40 a 120 °C



Usando el Sensor de Presión Piezoresistivo

$$r = r_0\left(1+\alpha T+\frac{2k\sigma}{1-k\sigma}\right)$$



$V_R = 5\text{V}$
$R = 10\text{K}\Omega$

```
Strain Transducer in Wheatstone Bridge
*
VR vcc 0 5V
R1 vcc outn 10K
R2 vcc outp 10K
R3 outn 0 10K
vk k 0 DC 0.006
vT T 0 DC 0
valfa alfa 0 DC 5e-3
vsigma sigma 0 DC 0
vr0 r0 0 DC 10K
bp outp 0 i=v(outp)/(v(r0)*(1+v(alfa)*v(T)
+ +2*v(k)*v(sigma)/(1-v(k)*v(sigma))))
.control
DC vsigma 0 100 0.1
plot v(outp,outn)
DC vT -40 120 0.1
plot v(outp,outn)
DC vsigma 0 100 1 vT -40 120 20
plot v(outp,outn)
.endc
.end
```

Usando el Sensor de Presión Piezoresistivo (cont)



$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

v_o *vs.* σ con $T = 0$ °C



Dr. J. E. Rayas-Sánchez 23

Usando el Sensor de Presión Piezoresistivo (cont)



$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

v_o *vs.* T con σ = 0 kPa



Dr. J. E. Rayas-Sánchez 24

Usando el Sensor de Presión Piezoresistivo (cont)



$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

v_o *vs.* σ con T : –40 a 120 °C



Dr. J. E. Rayas-Sánchez 25

Usando el Sensor de Presión Piezoresistivo (cont)



$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

v_o *vs.* σ con $T = 0$ °C



(controlling labels and units)

Dr. J. E. Rayas-Sánchez 26

Usando el Sensor de Presión Piezoresistivo (cont)



$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

v_o *vs.* T con $\sigma = 0$ kPa



(controlling labels and units)

Dr. J. E. Rayas-Sánchez 27

Usando el Sensor de Presión Piezoresistivo (cont)



$$r = r_0\left(1 + \alpha T + \frac{2k\sigma}{1 - k\sigma}\right)$$

v_o *vs.* σ con T : –40 a 120 °C



(controlling labels and units)

Dr. J. E. Rayas-Sánchez 28